"DG2022-10011 8:30:07.34 Thu 05/05/2022" **(1)**

Run this first: Define the gradient, gradient descent.

```
> myGradf:=proc(f,g,Gamma)
  return RaiseLowerIndices(InverseMetric(g),CovariantDerivative
  (f,Gamma),[1]);
  end:
> myGradDesPath:=proc(f,p0,eta,s,g,Gamma) local grad, ics,
  gradcomps, pt, A, i, gp, nsol, newthing, l1, l2, B, Q, Bv,
  vars, C, V, Ve, mystuff, Ics;
  Q:=DGinformation(g,"ObjectFrame");
  Bv:=DGinformation(Q,"FrameBaseVectors");
  vars:=DGinformation(Q,"FrameIndependentVariables");
  C:=[seq(vars[i](t),i=1..nops(vars))];
  V:=GeodesicEquations(C,Gamma,t);
  Ve:=DGinformation(V,"CoefficientList",[seq([i],i=1..nops(vars))
  ]);
  grad:=myGradf(f,g,Gamma);
       #Symbolically compute the gradient.
  gradcomps:=DGinformation(grad,"CoefficientList",[seq([j],j=1..
  nops(vars))]);              #Peel off the components
  pt:=p0;
            #This is the starting point.
  A:=Array([]);
            #Initialize the array that will store the path.
  ArrayTools:-Append(A,pt);
            #Append the original point to the path.
  for i in seq(j,j=1..s) do
            #s is the number of updates.
  mystuff:=ListTools:-FlattenOnce(convert(vars=pt,
  listofequations));
  gp:=eval(gradcomps,mystuff);              #evaluate the
  gradient at the point.
  Ics:=[seq([vars[j](0)=pt[j],D(vars[j])(0)=-eta*eval(gp[j],t=0)
  ], j=1..nops(vars))];  #Define the IC's.
  ics:=ListTools:-FlattenOnce(Ics);
  nsol:=dsolve(convert(ListTools:-FlattenOnce([Ve,ics]),set),{seq
  (vars[j](t),j=1..nops(vars))},numeric,maxfun=500000); #Obtain
  the numerical sol.
  newthing:=nsol(1);                #numerical sol includes
  phi' and theta' at 1
  l1:=[seq(lhs(newthing[j]),j=1..nops(newthing))];
```

```
l2:=[seq(rhs(newthing[j]),j=1..nops(newthing))];
pt:=[seq(l2[ListTools:-Search(vars[j](t),l1)],j=1..nops(vars))]
;        #This is the new point.
ArrayTools:-Append(A,pt);
            #Now we're appending the new point to the grad
path
od;
B:=[seq(A[j],j=op(2,A))];
          #Grad path is returned as a list.
end:
```

Example 1: Example from Wikipedia.org (nonlinear programming)

This is a 3-d example. We start by initializing a coordinate environment

```
> DGEnvironment[Manifold]([x,y,z],R3);
```

$$\text{Manifold: R3} \tag{2.1}$$

Here is the Euclidean metric (in Cartesian coordinates):

```
R3 > gR3:=evalDG(dx &t dx + dy &t dy + dz &t dz);
```

$$gR3 := dx \otimes dx + dy \otimes dy + dz \otimes dz \tag{2.2}$$

Here is the function to minimuze:

```
R3 > F:=-x*y-y*z;
```

$$F := -x\, y - y\, z \tag{2.3}$$

We require f1 <= 10 and f2 <= 2:

```
R3 > f1:=x^2+y^2+z^2;
```

$$f1 := x^2 + y^2 + z^2 \tag{2.4}$$

```
R3 > f2:=x^2-y^2+z^2;
```

$$f2 := x^2 - y^2 + z^2 \tag{2.5}$$

For the function G = (f1,f2,0), here is the Jacobian. The third row is needed for compatibility with the "invariant tensors" command.

```
R3 > J:=VectorCalculus:-Jacobian(Vector([f1,f2,0]),[x,y,z]);
```

$$J := \begin{bmatrix} 2\,x & 2\,y & 2\,z \\ 2\,x & -2\,y & 2\,z \\ 0 & 0 & 0 \end{bmatrix} \tag{2.6}$$

Here is a fancy way of calculating the nullspace of J:

```
R3 > X1:=LieAlgebras:-InvariantTensors([J],[D_x,D_y,D_z]);
```

$$X1 := \left[\partial_x - \frac{x}{z}\,\partial_z\right] \tag{2.7}$$

The nullspace is given over the space of smooth functions, so we can multiply by a smooth function to get a better vector field.

```
R3 > X:=evalDG(z*X1[1]);
```

$$X := z\,\partial_x - x\,\partial_z \tag{2.8}$$

Check that our vector field works:

```
R3 > LieDerivative(X,f1);
     LieDerivative(X,f2);
```

$$0$$

$$0 \tag{2.9}$$

We can calculate the flow, which is a rotation in the x-z plane.

```
R3 > FlX:=Flow(X,t);
```

$$FlX := x = z\sin(t) + x\cos(t),\;\; y = y,\;\; z = z\cos(t) - x\sin(t) \tag{2.10}$$

We obtain a coordinate from the vector field.

```
R3 > solve(op(1,FlX)[2][3][1],{t});
```

$$\left\{t = \arctan\left(\frac{z}{x}\right)\right\} \tag{2.11}$$

Equivalently, we solve X(f) = 1 for f:

```
R3 > pdsolve(LieDerivative(X,f(x,y,z))=1);
```

$$f(x, y, z) = \arctan\left(\frac{x}{z}\right) + \_F1(y, x^2 + z^2) \tag{2.12}$$

Now that we have identified coordinates, we need to initialize another coordinate environment.

```
R3 > DGEnvironment[Manifold]([t,u,v],M);
```

Manifold: M **(2.13)**

```
M > Phi:=Transformation(R3,M,[t=arctan(z/x),u=f1,v=f2]);
```

$$\Phi := t = \arctan\left(\frac{z}{x}\right),\ u = x^2 + y^2 + z^2,\ v = x^2 - y^2 + z^2 \quad \textbf{(2.14)}$$

In order to get the metric and loss function in these coordinates, we need to pullback by the inverse.

```
R3 > Phiin:=InverseTransformation(Phi);
```

$$Phiin := x = RootOf\left(\left(2\tan(t)^2 + 2\right)\_Z^2 - u - v\right),\ y = RootOf\left(2\_Z^2 - u + v\right),\ z = \tan(t)\, RootOf\left(\left(2\tan(t)^2 + 2\right)\_Z^2 - u - v\right) \quad \textbf{(2.15)}$$

Here is the new metric.

```
M > g:=Pullback(Phiin,gR3);
```

$$g := \left(\frac{u}{2} + \frac{v}{2}\right) dt \otimes dt + \frac{u}{4\left(u^2 - v^2\right)}\, du \otimes du - \frac{v}{4\left(u^2 - v^2\right)}\, du \otimes dv - \frac{v}{4\left(u^2 - v^2\right)}\, dv \otimes du + \frac{u}{4\left(u^2 - v^2\right)}\, dv \otimes dv \quad \textbf{(2.16)}$$

To get the function, we need to be a bit more explicit about the "inverse" function.

```
M > n1:=seq(ListTools:-FlattenOnce(op(1,Phiin)[2])[i],i=[1,3,5]
    );
```

$$n1 := RootOf\left(\left(2\tan(t)^2 + 2\right)\_Z^2 - u - v\right),\ RootOf\left(2\_Z^2 - u + v\right),\ \tan(t)\, RootOf\left(\left(2\tan(t)^2 + 2\right)\_Z^2 - u - v\right) \quad \textbf{(2.17)}$$

```
M > n2:=[x,y,z];
```

$$n2 := [x, y, z] \quad \textbf{(2.18)}$$

Here is the "inverse" function.

```
M > n3:=convert([seq(n2[i]=n1[i],i=1..3)],radical)
```

$$n3 := \left[x = \sqrt{-\frac{-u - v}{2\tan(t)^2 + 2}},\ y = \sqrt{\frac{1}{2}u - \frac{1}{2}v},\ z = \tan(t)\sqrt{-\frac{-u - v}{2\tan(t)^2 + 2}}\right] \quad \textbf{(2.19)}$$

And here is the new loss function in our new coordinates.

```
M > Fm:=simplify(eval(F,n3))
```

$$Fm := -\frac{1}{4}\,\frac{\sqrt{2}\,\sqrt{(u+v)\cos(t)^2}\,\sqrt{2\,u-2\,v}\,(\sin(t)+\cos(t))}{\cos(t)} \qquad \textbf{(2.20)}$$

First, let's optimize on the boundary.

The boundary has dimension 1.

```
M > DGEnvironment[Manifold]([tau],R);
```

Manifold: R **(2.21)**

Here is the boundary.

```
R > Theta:=Transformation(R,M,[t=tau,u=10,v=2]);
```

$$\Theta := t=\tau,\ u=10,\ v=2 \qquad \textbf{(2.22)}$$

Now we obtain the metric and the Christoffel symbols.

```
R > gR:=Pullback(Theta,g);
```

$$gR := 6\,d\tau \otimes d\tau \qquad \textbf{(2.23)}$$

```
R > Gammar:=Christoffel(gR);
```

$$Gammar := \nabla_{\partial_\tau}\partial_\tau = 0\,\partial_\tau \qquad \textbf{(2.24)}$$

Lastly, we get the loss function.

```
R > Fr:=simplify(eval(Fm,[t=tau,u=10,v=2]));
```

$$Fr := -2\sqrt{2}\,\sqrt{3}\,\mathrm{csgn}(\cos(\tau))\,(\sin(\tau)+\cos(\tau)) \qquad \textbf{(2.25)}$$

This is really a piecewise function.

if cos(tau) = 0, the function is 0. If cos(tau) is positive,

Here is the cos(tau) positive piece:

```
R > Frp:=simplify(Fr,useassumptions) assuming cos(tau)
    ::positive;
```

$$Frp := -2\sqrt{2}\,\sqrt{3}\,(\sin(\tau)+\cos(\tau)) \qquad \textbf{(2.26)}$$

Here is the cos(tau) negative piece.

```
R > Frn:=simplify(Fr,useassumptions) assuming cos(tau)
    ::negative;
```

$$Frn := 2\sqrt{2}\sqrt{3}\,(\sin(\tau)+\cos(\tau)) \quad (2.27)$$

This function is not continuous everywhere.

```
R > limit(Frp,tau=Pi/2)
```

$$-2\sqrt{2}\sqrt{3} \quad (2.28)$$

```
R > eval(Fr,tau=Pi/2);
```

$$0 \quad (2.29)$$

Let's work with the positive piece first. Here's the gradient:

```
R > myGradf(Frp,gR,Gammar);
```

$$-\frac{\sqrt{2}\sqrt{3}\,(\cos(\tau)-\sin(\tau))}{3}\,\partial_{\tau} \quad (2.30)$$

Obviously, the gradient is zero at pi/4. Here is the value at that point:

```
R > eval(Frp,tau=Pi/4);
```

$$-4\sqrt{3} \quad (2.31)$$

Decimal approximation for later comparison:

```
R > evalf(%);
```

$$-6.928203232 \quad (2.32)$$

Now we repeat the process for the negative piece.

```
R > myGradf(Frn,gR,Gammar);
```

$$\frac{\sqrt{2}\sqrt{3}\,(\cos(\tau)-\sin(\tau))}{3}\,\partial_{\tau} \quad (2.33)$$

```
R > eval(Frn,tau=Pi/4);
```

$$4\sqrt{3} \quad (2.34)$$

```
R > evalf(%);
```

$$6.928203232 \quad (2.35)$$

Now let's optimize on the interior.

We have to introduce new coordinates to force the inequalities to be satisfied.

```
M > DGEnvironment[Manifold]([a,b,c],M1);
```

Manifold: M1 **(2.36)**

```
M1 > eqns:= [t=a,u=-exp(b)+10,v=-exp(c)+2]; #[t=a,u=-b^2+10,
     v=-c^2+2];
```

$$eqns := [t = a,\ u = -e^{b} + 10,\ v = -e^{c} + 2] \tag{2.37}$$

```
M1 > Phi1:=Transformation(M1,M,eqns);
```

$$\Phi 1 := t = a,\ \ u = -e^{b} + 10,\ \ v = -e^{c} + 2 \tag{2.38}$$

Here's the metric in our new coordinates

```
M1 > g1:=Pullback(Phi1,g);
```

$$\begin{aligned} g1 := & -\left(\frac{e^{b}}{2} - 6 + \frac{e^{c}}{2}\right) da \otimes da + \frac{(e^{b} - 10)\, e^{2b}}{4\,(-e^{2b} + e^{2c} + 20\, e^{b} - 4\, e^{c} - 96)}\, db \otimes db \\ & - \frac{(e^{c} - 2)\, e^{b+c}}{4\,(-e^{2b} + e^{2c} + 20\, e^{b} - 4\, e^{c} - 96)}\, db \otimes dc \\ & - \frac{(e^{c} - 2)\, e^{b+c}}{4\,(-e^{2b} + e^{2c} + 20\, e^{b} - 4\, e^{c} - 96)}\, dc \otimes db \\ & + \frac{(e^{b} - 10)\, e^{2c}}{4\,(-e^{2b} + e^{2c} + 20\, e^{b} - 4\, e^{c} - 96)}\, dc \otimes dc \end{aligned} \tag{2.39}$$

Despite appearances, the inverse metric can be computed easily.

```
M1 > InverseMetric(g1)
```

$$\begin{aligned} & -\frac{2}{e^{b} - 12 + e^{c}}\, \partial_a \otimes \partial_a \\ & - \frac{4\,(-e^{2b} + e^{2c} + 20\, e^{b} - 4\, e^{c} - 96)\,(e^{b} - 10)\, e^{2c}}{e^{4c+2b} - e^{4b+2c} - 4\, e^{3c+2b} + 20\, e^{3b+2c} - 96\, e^{2b+2c}}\, \partial_b \otimes \partial_b \\ & - \frac{4\,(-e^{2b} + e^{2c} + 20\, e^{b} - 4\, e^{c} - 96)\,(e^{c} - 2)\, e^{b+c}}{e^{4c+2b} - e^{4b+2c} - 4\, e^{3c+2b} + 20\, e^{3b+2c} - 96\, e^{2b+2c}}\, \partial_b \otimes \partial_c \\ & - \frac{4\,(-e^{2b} + e^{2c} + 20\, e^{b} - 4\, e^{c} - 96)\,(e^{c} - 2)\, e^{b+c}}{e^{4c+2b} - e^{4b+2c} - 4\, e^{3c+2b} + 20\, e^{3b+2c} - 96\, e^{2b+2c}}\, \partial_c \otimes \partial_b \\ & - \frac{4\,(-e^{2b} + e^{2c} + 20\, e^{b} - 4\, e^{c} - 96)\,(e^{b} - 10)\, e^{2b}}{e^{4c+2b} - e^{4b+2c} - 4\, e^{3c+2b} + 20\, e^{3b+2c} - 96\, e^{2b+2c}}\, \partial_c \otimes \partial_c \end{aligned} \tag{2.40}$$

The Christoffel symbols are rather large, however.

```
M1 > Gamma1:=Christoffel(g1);
```

(2.41)

$$
\begin{aligned}
\mathit{\Gamma 1} := {} & \nabla_{\partial_a} \partial_a = \frac{\mathrm{e}^{b+2c} \left(\mathrm{e}^{2b} - \mathrm{e}^{2c} - 20\,\mathrm{e}^{b} + 4\,\mathrm{e}^{c} + 96\right) \left(\mathrm{e}^{b} - 12 + \mathrm{e}^{c}\right)}{\mathrm{e}^{4c+2b} - \mathrm{e}^{4b+2c} - 4\,\mathrm{e}^{3c+2b} + 20\,\mathrm{e}^{3b+2c} - 96\,\mathrm{e}^{2b+2c}}\, \partial_b \qquad \textbf{(2.41)} \\
& + \frac{\left(\mathrm{e}^{2b} - \mathrm{e}^{2c} - 20\,\mathrm{e}^{b} + 4\,\mathrm{e}^{c} + 96\right) \mathrm{e}^{2b+c} \left(\mathrm{e}^{b} - 12 + \mathrm{e}^{c}\right)}{\mathrm{e}^{4c+2b} - \mathrm{e}^{4b+2c} - 4\,\mathrm{e}^{3c+2b} + 20\,\mathrm{e}^{3b+2c} - 96\,\mathrm{e}^{2b+2c}}\, \partial_c \\
& , \nabla_{\partial_a} \partial_b = \frac{\mathrm{e}^{b}}{2\left(\mathrm{e}^{b} - 12 + \mathrm{e}^{c}\right)}\, \partial_a, \nabla_{\partial_a} \partial_c = \frac{\mathrm{e}^{c}}{2\left(\mathrm{e}^{b} - 12 + \mathrm{e}^{c}\right)}\, \partial_a \\
& , \nabla_{\partial_b} \partial_a = \frac{\mathrm{e}^{b}}{2\left(\mathrm{e}^{b} - 12 + \mathrm{e}^{c}\right)}\, \partial_a, \nabla_{\partial_b} \partial_b = \big(2\,\mathrm{e}^{6c+2b} + 70\,\mathrm{e}^{3b+4c} \\
& + 888\,\mathrm{e}^{4b+2c} - 6720\,\mathrm{e}^{3b+2c} + 12\,\mathrm{e}^{4b+3c} - 3\,\mathrm{e}^{4b+4c} - 280\,\mathrm{e}^{3b+3c} \\
& + \mathrm{e}^{6b+2c} - 50\,\mathrm{e}^{5b+2c} + 18432\,\mathrm{e}^{2b+2c} - 16\,\mathrm{e}^{5c+2b} - 352\,\mathrm{e}^{4c+2b} \\
& + 1536\,\mathrm{e}^{3c+2b}\big) \big/ \big(2\left(-\mathrm{e}^{2b} + \mathrm{e}^{2c} + 20\,\mathrm{e}^{b} - 4\,\mathrm{e}^{c} - 96\right) \big(\mathrm{e}^{4c+2b} - \mathrm{e}^{4b+2c} \\
& - 4\,\mathrm{e}^{3c+2b} + 20\,\mathrm{e}^{3b+2c} - 96\,\mathrm{e}^{2b+2c}\big)\big)\, \partial_b - \big(\left(\mathrm{e}^{c} - 2\right) \big(-4\,\mathrm{e}^{4b+2c} \\
& + \mathrm{e}^{4b+3c} + 20\,\mathrm{e}^{5b+c} - \mathrm{e}^{6b+c} - 96\,\mathrm{e}^{4b+c}\big)\big) \big/ \big(2\big(-\mathrm{e}^{2b} + \mathrm{e}^{2c} + 20\,\mathrm{e}^{b} - 4\,\mathrm{e}^{c} \\
& - 96\big) \left(\mathrm{e}^{4c+2b} - \mathrm{e}^{4b+2c} - 4\,\mathrm{e}^{3c+2b} + 20\,\mathrm{e}^{3b+2c} - 96\,\mathrm{e}^{2b+2c}\right)\big)\, \partial_c \\
& , \nabla_{\partial_b} \partial_c = \big(\left(\mathrm{e}^{c} - 2\right) \big(-20\,\mathrm{e}^{3b+3c} + \mathrm{e}^{4b+3c} + 96\,\mathrm{e}^{3c+2b} - \mathrm{e}^{5c+2b} \\
& + 4\,\mathrm{e}^{4c+2b}\big)\big) \big/ \big(2\left(-\mathrm{e}^{2b} + \mathrm{e}^{2c} + 20\,\mathrm{e}^{b} - 4\,\mathrm{e}^{c} - 96\right) \big(\mathrm{e}^{4c+2b} - \mathrm{e}^{4b+2c} \\
& - 4\,\mathrm{e}^{3c+2b} + 20\,\mathrm{e}^{3b+2c} - 96\,\mathrm{e}^{2b+2c}\big)\big)\, \partial_b - \big(-\mathrm{e}^{4b+4c} + 10\,\mathrm{e}^{3b+4c} \\
& + \mathrm{e}^{6b+2c} - 30\,\mathrm{e}^{5b+2c} + 4\,\mathrm{e}^{4b+3c} + 296\,\mathrm{e}^{4b+2c} - 40\,\mathrm{e}^{3b+3c} \\
& - 960\,\mathrm{e}^{3b+2c}\big) \big/ \big(2\left(-\mathrm{e}^{2b} + \mathrm{e}^{2c} + 20\,\mathrm{e}^{b} - 4\,\mathrm{e}^{c} - 96\right) \big(\mathrm{e}^{4c+2b} - \mathrm{e}^{4b+2c} \\
& - 4\,\mathrm{e}^{3c+2b} + 20\,\mathrm{e}^{3b+2c} - 96\,\mathrm{e}^{2b+2c}\big)\big)\, \partial_c, \nabla_{\partial_c} \partial_a = \frac{\mathrm{e}^{c}}{2\left(\mathrm{e}^{b} - 12 + \mathrm{e}^{c}\right)}\, \partial_a \\
& , \nabla_{\partial_c} \partial_b = \big(\left(\mathrm{e}^{c} - 2\right) \big(-20\,\mathrm{e}^{3b+3c} + \mathrm{e}^{4b+3c} + 96\,\mathrm{e}^{3c+2b} - \mathrm{e}^{5c+2b} \\
& + 4\,\mathrm{e}^{4c+2b}\big)\big) \big/ \big(2\left(-\mathrm{e}^{2b} + \mathrm{e}^{2c} + 20\,\mathrm{e}^{b} - 4\,\mathrm{e}^{c} - 96\right) \big(\mathrm{e}^{4c+2b} - \mathrm{e}^{4b+2c} \\
& - 4\,\mathrm{e}^{3c+2b} + 20\,\mathrm{e}^{3b+2c} - 96\,\mathrm{e}^{2b+2c}\big)\big)\, \partial_b - \big(-\mathrm{e}^{4b+4c} + 10\,\mathrm{e}^{3b+4c} \\
& + \mathrm{e}^{6b+2c} - 30\,\mathrm{e}^{5b+2c} + 4\,\mathrm{e}^{4b+3c} + 296\,\mathrm{e}^{4b+2c} - 40\,\mathrm{e}^{3b+3c} \\
& - 960\,\mathrm{e}^{3b+2c}\big) \big/ \big(2\left(-\mathrm{e}^{2b} + \mathrm{e}^{2c} + 20\,\mathrm{e}^{b} - 4\,\mathrm{e}^{c} - 96\right) \big(\mathrm{e}^{4c+2b} - \mathrm{e}^{4b+2c} \\
& - 4\,\mathrm{e}^{3c+2b} + 20\,\mathrm{e}^{3b+2c} - 96\,\mathrm{e}^{2b+2c}\big)\big)\, \partial_c, \nabla_{\partial_c} \partial_c = -\big(\big(\mathrm{e}^{b} \\
& - 10\big) \left(\mathrm{e}^{3b+4c} - 20\,\mathrm{e}^{4c+2b} - \mathrm{e}^{6c+b} + 96\,\mathrm{e}^{4c+b} + 4\,\mathrm{e}^{5c+b}\right)\big) \big/ \big(2\big(-\mathrm{e}^{2b} \\
& + \mathrm{e}^{2c} + 20\,\mathrm{e}^{b} - 4\,\mathrm{e}^{c} - 96\big) \big(\mathrm{e}^{4c+2b} - \mathrm{e}^{4b+2c} - 4\,\mathrm{e}^{3c+2b} + 20\,\mathrm{e}^{3b+2c} \\
& - 96\,\mathrm{e}^{2b+2c}\big)\big)\, \partial_b + \big(2\,\mathrm{e}^{6b+2c} - 80\,\mathrm{e}^{5b+2c} + 14\,\mathrm{e}^{4b+3c} \\
& + 1344\,\mathrm{e}^{3c+2b} - 3\,\mathrm{e}^{4b+4c} + 60\,\mathrm{e}^{3b+4c} + \mathrm{e}^{6c+2b} - 10\,\mathrm{e}^{5c+2b}
\end{aligned}
$$

$$- 280\, e^{3 b + 3 c} - 264\, e^{4 c + 2 b} + 1184\, e^{4 b + 2 c} + 18432\, e^{2 b + 2 c} - 7680\, e^{3 b + 2 c}) \Big/ (2\, (-e^{2 b} + e^{2 c} + 20\, e^{b} - 4\, e^{c} - 96)\, (e^{4 c + 2 b} - e^{4 b + 2 c} - 4\, e^{3 c + 2 b} + 20\, e^{3 b + 2 c} - 96\, e^{2 b + 2 c}))\, \partial_c$$

This is still a flat space, though:

```
M1 > CurvatureTensor(Gamma1);
```

$$0\, \partial_a \otimes da \otimes da \otimes da \tag{2.42}$$

Here is the loss function in our new coordinates.

```
M1 > Fm1:=eval(Fm,eqns);
```

$$Fm1 := -\frac{1}{4}\, \frac{\sqrt{2}\, \sqrt{(-e^{b} + 12 - e^{c})\, \cos(a)^2}\, \sqrt{-2\, e^{b} + 16 + 2\, e^{c}}\, (\sin(a) + \cos(a))}{\cos(a)} \tag{2.43}$$

We now initiate gradient descent. Here is the starting point.

```
M1 > mypoint:=[1,1,1];
```

$$mypoint := [1, 1, 1] \tag{2.44}$$

Now we optimize in our choice of coordinates, tracking the path made.

```
M1 > Mpath:=myGradDesPath(Fm1,mypoint,0.001,114,g1,Gamma1):
```

Now we get the path in (t,u,v) coordinates.

```
M1 > evals1:=[seq(rhs(eqns[i]),i=1..nops(eqns))]:
M1 > Rpath1:=[seq(evalf(eval(evals1,ListTools:-FlattenOnce
     (convert([a, b, c]=Mpath[i],listofequations)))),i=1..
     nops(Mpath))]:
```

Now we get the path in the original coordinates.

```
M1 > evals2:=[seq(rhs(n3[i]),i=1..nops(n3))]:
M1 > Rpath2:=[seq(evalf(eval(evals2,ListTools:-FlattenOnce
     (convert([t, u, v]=Rpath1[i],listofequations)))),i=1..
     nops(Rpath1))]:
```

We will now plot the loss function values as the optimization process took place.

```
M1 > myloss:=proc(p,F)
```

```
    return eval(F,ListTools:-FlattenOnce(convert([x,y,z]=p,
    listofequations)));
    end:
M1 > losses:=[seq(evalf(myloss(Rpath2[i],F)), i=1..nops
    (Rpath2))]:
M1 > plot([seq(i,i=1..nops(Rpath2))],losses);
```



Here is the approximated minimum (on the interior):

```
M1 > losses[-1];
```

$$-6.92186603302018 \tag{2.45}$$

compare with the minimum obtained analytically on the boundary:

```
M1 > evalf(-4*sqrt(3));
```

$$-6.928203232 \tag{2.46}$$

Here is the last point obtained in (a,b,c) coordinates:

```
M1 > Mpath[-1]
```
$$[0.965681551410841, -3.73550959315653, 0.999927275134812] \tag{2.47}$$

Here is the last point obtained in (x,y,z) coordinates.

```
M1 > Rpath2[-1];
```
$$[1.22390464464707, 2.31238222839611, 1.76948716922327] \tag{2.48}$$

Truthfully, the original wikipedia article wanted the maximum value, not the minimum: we minimized the negative function. Here is the requested maximum (on the interior):

```
M1 > eval(-F,[x=Rpath2[-1][1],y=Rpath2[-1][2],z=Rpath2[-1][3]
     ])
```
$$6.92186603302018 \tag{2.49}$$

Example 2: Positive Definite Matrices of size 2x2.

In this example, we will look to optimize the components of a positive definite matrix used in a function.

```
> DGEnvironment[Manifold]([w,x,y,z],R4);
```
$$\text{Manifold: R4} \tag{3.1}$$

```
R4 > gR4:=evalDG(dw &t dw + dx &t dx + dy &t dy + dz &t dz);
```
$$gR4 := dw \otimes dw + dx \otimes dx + dy \otimes dy + dz \otimes dz \tag{3.2}$$

Below is a function, given some positive definite matrix A and a vector b with positive components.

```
R4 > F:=proc(w,x,y,z,A,b) local S, term;
     S:=Matrix([[w,x],[y,z]]);
     term:=LinearAlgebra:-Trace(A.S^(-1)) + (b^+).S.b;
     return term;
     end;
```
$F := \textbf{proc}(w, x, y, z, A, b)$ (3.3)

$\quad \textbf{local}\ S, term;$

$\quad S := Matrix([[w, x], [y, z]]);$

$\quad term := LinearAlgebra{:-}Trace(A\ .\ (1\ /\ S)) + b\text{^}\ \text{`\%T`}\ .\ S\ .\ b;$

```
    return term
end proc
```

In this example, we make the following arbitrary choice for A:

```
R4 > myA:=Matrix([[1,-2],[-2,6]]);
```

$$myA := \begin{bmatrix} 1 & -2 \\ -2 & 6 \end{bmatrix} \quad (3.4)$$

```
R4 > LinearAlgebra:-IsDefinite(myA,'query'=
     'positive_definite');
```

$$true \quad (3.5)$$

We make the following choice for b:

```
R4 > myb:=Vector([1,4]);
```

$$myb := \begin{bmatrix} 1 \\ 4 \end{bmatrix} \quad (3.6)$$

To be clear, we will examine a function of the variables w, x, y, and z, which are the components of S:

```
R4 > S=Matrix([[w,x],[y,z]]);
```

$$S = \begin{bmatrix} w & x \\ y & z \end{bmatrix} \quad (3.7)$$

Thus, our scalar-valued function of the components of the matrix S become:

```
R4 > myF:=F(w,x,y,z,myA,myb);
```

$$myF := \frac{z}{w\,z - x\,y} + \frac{2\,y}{w\,z - x\,y} + \frac{2\,x}{w\,z - x\,y} + \frac{6\,w}{w\,z - x\,y} + w + 4\,y + 4\,x + 16\,z \quad (3.8)$$

For a matrix S=Matrix([[w,x],[y,z]]) to be positive definite, we need the following: y=x, w>0, z - x^2/w > 0. Try: tau = z-x^2/w, w=exp(beta), x=gamma, y= gamma, so that z = tau + gamma^2/exp(beta).

additional bounds: tau > 0: use tau = exp(alpha).

```
R4 > DGEnvironment[Manifold]([alpha, beta, gamma1],M);
```

Manifold: M **(3.9)**

gamma is a protected symbol in Maple, so we'll use gamma1.

```
M > eq1:=[w=exp(beta), x=gamma1, y=gamma1, z=exp(alpha) +
    (gamma1)^2/exp(beta)];
```

$$eq1 := \left[w = e^{\beta}, x = \gamma 1, y = \gamma 1, z = e^{\alpha} + \frac{\gamma 1^{2}}{e^{\beta}}\right] \tag{3.10}$$

```
M > Phi:=Transformation(M,R4,eq1):
```

Here's the metric on the manifold of positive definite matrices (in our coordinates)
.

```
M > g:=Pullback(Phi,gR4);
```

$$g := e^{2\alpha}\, d\alpha \otimes d\alpha - e^{\alpha-\beta}\gamma 1^{2}\, d\alpha \otimes d\beta + 2\,\gamma 1\, e^{\alpha-\beta}\, d\alpha \otimes d\gamma 1 - e^{\alpha-\beta}\gamma 1^{2}\, d\beta \otimes d\alpha + \left(e^{2\beta} + \gamma 1^{4} e^{-2\beta}\right) d\beta \otimes d\beta - 2\,\gamma 1^{3} e^{-2\beta}\, d\beta \otimes d\gamma 1 + 2\,\gamma 1\, e^{\alpha-\beta}\, d\gamma 1 \otimes d\alpha - 2\,\gamma 1^{3} e^{-2\beta}\, d\gamma 1 \otimes d\beta + \left(2 + 4\,\gamma 1^{2} e^{-2\beta}\right) d\gamma 1 \otimes d\gamma 1 \tag{3.11}$$

Here is the metric connection.

```
M > Gamma:=Christoffel(g);
```

$$\Gamma := \nabla_{\partial_{\alpha}} \partial_{\alpha} = \partial_{\alpha}, \nabla_{\partial_{\beta}} \partial_{\beta} = 2\,\gamma 1^{2} e^{-\alpha-\beta}\partial_{\alpha} + \partial_{\beta}, \nabla_{\partial_{\beta}} \partial_{\gamma 1} = -2\, e^{-\alpha-\beta}\gamma 1\, \partial_{\alpha}, \nabla_{\partial_{\gamma 1}} \partial_{\beta} = -2\, e^{-\alpha-\beta}\gamma 1\, \partial_{\alpha}, \nabla_{\partial_{\gamma 1}} \partial_{\gamma 1} = 2\, e^{-\alpha-\beta}\partial_{\alpha} \tag{3.12}$$

Just for fun, we ask: how many Killing vectors? Is this a space of constant curvature?

```
M > kv:=KillingVectors(g):
    nops(kv);
    CurvatureTensor(Gamma);
```

$$6$$

$$0\, \partial_{\alpha} \otimes d\alpha \otimes d\alpha \otimes d\alpha \tag{3.13}$$

This is a flat space!

Anyway, we proceed by writing our function in our new coordinate system. The result is surprisingly simple.

```
M > F1:=simplify(eval(myF,eq1));
```

$$F1 := e^{-\beta} + 4\, e^{-\alpha-\beta}\gamma 1 + e^{-\alpha-2\beta}\gamma 1^{2} + e^{\beta} + 16\, e^{\alpha} + 8\,\gamma 1 + 16\,\gamma 1^{2} e^{-\beta} + 6\, e^{-\alpha} \tag{3.14}$$

For optimization, we initialize the new weights as 0's. This corresponds to the identity matrix (see original coordinates).

```
M > mypoint:=[0,0,0];
```

$$mypoint := [0, 0, 0] \tag{3.15}$$

Now, we optimize and give the path in terms of our new coordinates and the original coordinates.

```
M > Mpath:=myGradDesPath(F1,mypoint,0.02,1000,g,Gamma):
M > evals1:=[seq(rhs(eq1[i]),i=1..nops(eq1))]:
M > Rpath1:=[seq(evalf(eval(evals1,ListTools:-FlattenOnce
    (convert([alpha, beta, gamma1]=Mpath[i],listofequations)))
    ),i=1..nops(Mpath))]:
```

What are the matrices which correspond to these coordinates?

```
M > Rpath2:=[seq(Matrix([[Rpath1[i][1],Rpath1[i][2]],[Rpath1[i]
    [3],Rpath1[i][4]]]),i=1..nops(Rpath1))]:
```

We ask: did we stay on the manifold of positive definite matrices?

```
M > TF:=[seq(LinearAlgebra:-IsDefinite(Rpath2[i], 'query'=
    'positive_definite') ,i=1..nops(Rpath2))]:
    convert(TF,set);
```

$$\{true\} \tag{3.16}$$

Let's see how we did: whether the function decreased or not.

```
M > myloss:=proc(p,F)
    return eval(F,ListTools:-FlattenOnce(convert([w,x,y,z]=p,
    listofequations)));
    end:
M > losses:=[seq(evalf(myloss(Rpath1[i],myF)), i=1..nops
    (Rpath1))]:
M > plot([seq(i,i=1..nops(Rpath1))],losses);
```




```
M > Mpath[-1];
```

$$[-0.598519139669302, 1.04632428676618, -0.901045512104919] \quad (3.17)$$

Here is where we ended up (given as an explicit matrix):

```
M > Rpath2[-1];
```

$$\begin{bmatrix} 2.84716649307296 & -0.901045512104919 \\ -0.901045512104919 & 0.834779689357431 \end{bmatrix} \quad (3.18)$$

Here's the loss function value.

```
M > losses[-1];
```

$$18.1420834467142 \quad (3.19)$$

Example 3: application to Gas Transmission Compressor Design

https://github.com/P-N-Suganthan/2020-RW-Constrained-Optimisation/blob/master/Problem-Definitions.pdf problem 40.

```
> DGEnvironment[Manifold]([x1,x2,x3,x4],R4);
```

Manifold: R4 (4.1)

Here is the objective function:

```
R4 > F:=8.61*10^(5)*(x1)^(1/2)*x2*(x3)^(-1/2)*(x4)^(-1/2) +
     3.69*10^(4)*x3 + 7.72*10^(8)*(x1)^(-1)*(x2)^(0.219) -
     765.43*10^(6)*(x1)^(-1);
```

$$F := \frac{8.6100000\ 10^{5} \sqrt{x1}\ x2}{\sqrt{x3}\ \sqrt{x4}} + 36900.00\ x3 + \frac{7.720000000\ 10^{8}\ x2^{0.219}}{x1} - \frac{7.654300000\ 10^{8}}{x1} \tag{4.2}$$

Given constraints:

x4/(x2)^2 + 1/(x2)^2 -1 <= 0

20 <= x1 <= 50
1 <= x2 <= 10
20 <= x3 <= 50
0.1 <= x4 <= 60

```
R4 > gR4:=evalDG(dx1 &t dx1 + dx2 &t dx2 + dx3 &t dx3 + dx4
     &t dx4);
```

$$gR4 := dx1 \otimes dx1 + dx2 \otimes dx2 + dx3 \otimes dx3 + dx4 \otimes dx4 \tag{4.3}$$

We will first address the following constraint: x4/(x2)^2 + 1/(x2)^2 -1 <= 0.

The failed attempt to use all constraints.

We'll introduce new variables for which the first four constraints are trivialized.

```
R4 > DGEnvironment[Manifold]([a1,a2,a3,a4],N1);
```

Manifold: N1 **(4.1.1)**

This is the function f1 for which f1 <= 0.

```
N1 > f1:=x4/(x2)^2 + 1/(x2)^2 -1;
```

$$f1 := \frac{x4}{x2^2} + \frac{1}{x2^2} - 1 \tag{4.1.2}$$

Below, we've defined new coordinates so that the first four constraints are sasisfied.

```
N1 > eqn1:=[x1=((50-20)/2)*sin(a1)+(50+20)/2, x2=((10-1)/2)
     *sin(a2)+(10+1)/2, x3=((50-20)/2)*sin(a3)+(50+20)/2,
     x4=2/Pi*((60-0.1)/2)*arctan(a4)+(60+0.1)/2];
```

$$eqn1 := \left[x1 = 15\sin(a1) + 35,\ x2 = \frac{9}{2}\sin(a2) + \frac{11}{2},\ x3 = 15\sin(a3) + 35,\ x4 = 19.06676218\arctan(a4) + 30.05000000\right] \tag{4.1.3}$$

Here's the metric on our new coordinates.

```
N1 > gn1:=Pullback(Transformation(N1,R4,eqn1),gR4);
```

$$gn1 := 225\cos(a1)^2\, da1 \otimes da1 + \frac{81\cos(a2)^2}{4}\, da2 \otimes da2 + 225\cos(a3)^2\, da3 \otimes da3 + \frac{363.54142}{(a4^2+1)^2}\, da4 \otimes da4 \tag{4.1.4}$$

Try to incorporate the final constraint.

Here's f1 in our new coordinates.

```
N1 > fn1:=Pullback(Transformation(N1,R4,eqn1),f1);
```

$$fn1 := \frac{76.26704872\arctan(a4) + 3.2 - 81.\sin(a2)^2 - 198.\sin(a2)}{(9.\sin(a2) + 11.)^2} \tag{4.1.1.1}$$

Here's the objective function in our new coordiantes.

```
N1 > Fn1:=simplify(Pullback(Transformation(N1,R4,eqn1),F)
     );
```

$$Fn1 := \Big(1.162350000\ 10^{7}\ \Big(13.28343442\ (4.5\sin(a2) + 5.5)^{\frac{219}{1000}}\ \sqrt{15.\sin(a3)+35.}\ \sqrt{19.06676218\arctan(a4)+30.05} + ((0.1428571429\sin(a3)+0.3333333333)\sin(a1) + 0.3333333333\sin(a3) - 12.39260980)\ \sqrt{15.\sin(a3)+35.}\ \sqrt{19.06676218\arctan(a4)+30.05} + ((1.000000000\sin(a2)+1.222222222)\sin(a1) + 2.333333333\sin(a2) + 2.851851852)\ \sqrt{15.\sin(a1)+35.}\Big)\Big) \Big/ \Big(\sqrt{19.06676218\arctan(a4)+30.05}\ \sqrt{15\sin(a3)+35}\ (3.\sin(a1) + 7.)\Big) \quad \textbf{(4.1.1.2)}$$

We compute the jacobian of f1, adding extra rows for compatibility with "invariant tensors."

```
N1 > Jm:=VectorCalculus:-Jacobian(Vector([fn1,0,0,0]),
     [a1,a2,a3,a4]);
```

$$Jm := \left[\left[0, \frac{-162.\sin(a2)\cos(a2) - 198.\cos(a2)}{(9.\sin(a2)+11.)^{2}} - \frac{1}{(9.\sin(a2)+11.)^{3}}\big(18.\,\big(76.26704872\arctan(a4) + 3.2 - 81.\sin(a2)^{2} - 198.\sin(a2)\big)\cos(a2)\big), 0, \frac{76.26704872}{(a4^{2}+1)\,(9.\sin(a2)+11.)^{2}}\right], \left[0, 0, 0, 0\right], \left[0, 0, 0, 0\right], \left[0, 0, 0, 0\right]\right] \quad \textbf{(4.1.1.3)}$$

Here is the nullspace.

```
N1 > Xsm:=LieAlgebras:-InvariantTensors([Jm],[D_a1,D_a2,
     D_a3,D_a4]);
```

(4.1.1.4)

$$Xsm := \left[\partial_{a1}, \partial_{a2} + \frac{\cos(a2)\ (1372.806877 \arctan(a4) + 2235.600000)\ (a4^2+1)}{686.4034385 \sin(a2) + 838.9375359} \partial_{a4}, \partial_{a3}\right] \tag{4.1.1.4}$$

The first and third vectors correspond to coordinates a1 and a3, respectively. But what about the second?

```
N1 > pdsolve(LieDerivative(Xsm[2],f(a1,a2,a3,a4))=1)
```

$$f(a1, a2, a3, a4) = a2 + \_F1(a1, a3, (2\ (1372806877 \arctan(a4) + 2235600000)) / (187878205866855085987 + 230339843731390876860 \sin(a2) - 47114968038462328225 \cos(2\ a2))) \tag{4.1.1.5}$$

It looks like we can take a2 as another coordinate. f1 will essentially be the fourth coordinate.

```
N1 > DGEnvironment[Manifold]([b1,b2,b3,b4],N2);
```

Manifold: N2 **(4.1.1.6)**

Now we just need to make -exp(b4) = f1, since it can then never be greater than 0.

```
N2 > eqn2:=[b1=a1,b2=a2,b3=a3,b4=ln(-fn1)];
```

$$eqn2 := \left[b1 = a1, b2 = a2, b3 = a3, b4 = \ln\left(-\frac{76.26704872 \arctan(a4) + 3.2 - 81. \sin(a2)^2 - 198. \sin(a2)}{(9. \sin(a2) + 11.)^2}\right)\right] \tag{4.1.1.7}$$

```
N2 > Phi2:=Transformation(N1,N2,eqn2);
```

$$\Phi 2 := b1 = a1,\ b2 = a2,\ b3 = a3,\ b4 = \ln\left(-\frac{76.26704872 \arctan(a4) + 3.2 - 81. \sin(a2)^2 - 198. \sin(a2)}{(9. \sin(a2) + 11.)^2}\right) \tag{4.1.1.8}$$

Here's the inverse transformation so that we can pullback the metric and loss function.

```
N1 > Phi2in:=InverseTransformation(Phi2);
```

$$Phi2in := a1 = b1,\ a2 = b2,\ a3 = b3,\ a4 = \tag{4.1.1.9}$$

$$-\tan\big(1.062057617\,e^{b4}\sin(b2)^2+2.596140841\,e^{b4}\sin(b2)-1.062057617\sin(b2)^2+1.586530514\,e^{b4}-2.596140841\sin(b2)+0.04195783177\big)$$

N2 > gn2:=simplify(Pullback(Phi2in,gn1));

$$gn2 := 225\cos(b1)^2\,db1\otimes db1-1.\,10^{-6}\Big(\big(\big((\cos(b2)^2-\sin(b2)-2.)\,e^{2\,b4}+(e^{b4}+1)\cos(b2)^2+(2.\sin(b2)-1)\,e^{b4}-\sin(b2)-2.\big)\cos\big((-2.596140841\sin(b2)+1.062057617\cos(b2)^2-2.648588131)\,e^{b4}+1.020099785-1.062057617\cos(b2)^2+2.596140841\sin(b2)\big)\big)^4+\big((-\cos(b2)^2+\sin(b2)+2.)\,e^{2\,b4}+(-e^{b4}-1)\cos(b2)^2+\sin(b2)+e^{b4}+2.\big)\cos\big((-2.596140841\sin(b2)+1.062057617\cos(b2)^2-2.648588131)\,e^{b4}+1.020099785-1.062057617\cos(b2)^2+2.596140841\sin(b2)\big)\big)^2+(1.640250001\,10^9\cos(b2)^2-4.009500001\,10^9\sin(b2)-4.090500001\,10^9)\,e^{2\,b4}+(-3.280500001\,10^9\,e^{b4}+1.640250001\,10^9)\cos(b2)^2+(8.019000001\,10^9\sin(b2)+8.181000000\,10^9)\,e^{b4}-4.110750001\,10^9-4.009500001\,10^9\sin(b2)\Big)\cos(b2)^2\,db2\otimes db2-820.1250004\big(\big((\sin(b2)+3.666666667)\,e^{b4}-\sin(b2)-3.666666667\big)\cos(b2)^2+(-5.481481479\sin(b2)-5.492455419)\,e^{b4}+5.481481479\sin(b2)+5.492455419\big)\,e^{b4}\cos(b2)\,db2\otimes db4+225\cos(b3)^2\,db3\otimes db3-820.1250004\big(\big((\sin(b2)+3.666666667)\,e^{b4}-\sin(b2)-3.666666667\big)\cos(b2)^2+(-5.481481479\sin(b2)-5.492455419)\,e^{b4}+5.481481479\sin(b2)+5.492455419\big)\,e^{b4}\cos(b2)\,db4\otimes db2+\big(410.0625002\cos(b2)^4+(-2004.75\sin(b2)-4495.499999)\cos(b2)^2+4999.500002\sin(b2)+5000.499999\big)\,e^{2\,b4}\,db4\otimes db4 \qquad \textbf{(4.1.1.10)}$$

N2 > Gamma2:=simplify(Christoffel(gn2)):

N2 > Fn2:=simplify(Pullback(Phi2in,Fn1)):

N2 > Fn2;

$$\Big(1.162350000\,10^7\Big(\Big(13.28343442\,(4.5\sin(b2)+5.5)^{\frac{219}{1000}}\sqrt{15.\sin(b3)+35.}+((0.1428571429\sin(b3)$$

$$+ 0.3333333333) \sin(b1) + 0.3333333333 \sin(b3)$$

$$- 12.39260980) \sqrt{15. \sin(b3) + 35.})$$

$$(19.06676218 \arctan((1. \sin((-2.596140841 \sin(b2) + 1.062057617 \cos(b2)^2 - 2$$

$$+ 2.596140841 \sin(b2)))/(\cos((-2.596140841 \sin(b2)$$

$$+ 1.062057617 \cos(b2)^2 - 2.648588131) e^{b4} + 1.020099785$$

$$- 1.062057617 \cos(b2)^2 + 2.596140841 \sin(b2)))) + 30.05)^{1/2}$$

$$+ ((1.000000000 \sin(b2) + 1.222222222) \sin(b1)$$

$$+ 2.333333333 \sin(b2) + 2.851851852) \sqrt{15. \sin(b1) + 35.}))$$

$$\Big/$$

$$\Big(\sqrt{15 \sin(b3) + 35}\ (19.06676218 \arctan((1. \sin(($$

$$-2.596140841 \sin(b2) + 1.062057617 \cos(b2)^2 - 2.648588131)$$

$$e^{b4} + 1.020099785 - 1.062057617 \cos(b2)^2$$

$$+ 2.596140841 \sin(b2)))/(\cos((-2.596140841 \sin(b2)$$

$$+ 1.062057617 \cos(b2)^2 - 2.648588131) e^{b4} + 1.020099785$$

$$- 1.062057617 \cos(b2)^2 + 2.596140841 \sin(b2)))) + 30.05)$$

$$^{1/2}\ (3. \sin(b1) + 7.)\Big)$$

Things are looking very dicey, but we technically have coordinates where all of the inequalities are satisfied.

The problem is that we cannot so much as compute the Riemannian gradient in a timely fashion: our ambitions have caught up to us!

N2 > #myGradf(Fn2,gn2,Gamma2):

Warning, computation interrupted

As a future project, perhaps there are different coordinates that can be introduced that make this approach feasible. For now, it seems difficult to introduce more constraints than variables.

Now, let's return to the original problem and try another approach.

R4 > DGEnvironment[Manifold]([y1,y2,y3,y4],M1);

$$\textit{Manifold: M1} \tag{4.4}$$

Let's deal with the more complicated constraint first.

M1 > eqn1:=[x1=y1,x2=y2,x3=y3,x4=y2^2-y4];

$$eqn1 := [x1 = y1, x2 = y2, x3 = y3, x4 = y2^2 - y4] \tag{4.5}$$

M1 > F1:=eval(F,eqn1);

$$F1 := \frac{8.6100000\,10^5 \sqrt{y1}\, y2}{\sqrt{y3}\,\sqrt{y2^2 - y4}} + 36900.00\, y3 + \frac{7.720000000\,10^8\, y2^{0.219}}{y1} - \frac{7.654300000\,10^8}{y1} \tag{4.6}$$

The constraint is equivalent to 1 <= (x2)^2 - x4. Thus, we set y4 = (x2)^2 - x4 and impose the constraint 1 <= y4.

As a sub problem (using the constraints for x2 and x4), we find that y4 <= 100 - 0.1.

As a consequence of 1 <= (x2)^2 - x4 and our other boundaries, we can't get it exactly. 1 <= y2 <= 100-0.1 will ignore the boundaries for x2.

We would just need to track whether the boundaries for x2 are ever approached or not.

M1 > Phi1:=Transformation(M1,R4,eqn1):

M1 > gM1:=Pullback(Phi1,gR4);

$$gM1 := dy1 \otimes dy1 + (4\, y2^2 + 1)\, dy2 \otimes dy2 - 2\, y2\, dy2 \otimes dy4 + dy3 \otimes dy3 - 2\, y2\, dy4 \otimes dy2 + dy4 \otimes dy4 \tag{4.7}$$

```
M1 > Gamma1:=Christoffel(gM1)
```

$$\Gamma 1 := \nabla_{\partial_{y2}} \partial_{y2} = -2\, \partial_{y4} \tag{4.8}$$

1<= y4 <= 100-0.1. (new inequality)
20<=y1 <=50 (old inequality)
1<=y2 <= 10 (old inequality)
20 <= y3 <= 50 (old inequality)

```
M1 > DGEnvironment[Manifold]([z1,z2,z3,z4],M2);
```

$$\text{Manifold: } M2 \tag{4.9}$$

These equations will guarantee that our inequality constraints will be satisfied.

```
M2 > eqn2:=[y1=15*sin(z1)+35, y2=(10-1)/2*sin(z2)+abs(10-(10
     -1)/2), y3=15*sin(z3)+35, y4=(100-0.1-1)/2*sin(z4)+abs(
     (100-0.1)-(100-0.1-1)/2)];
```

$$eqn2 := \left[y1 = 15 \sin(z1) + 35,\ y2 = \frac{9}{2} \sin(z2) + \frac{11}{2},\ y3 = 15 \sin(z3) + 35,\ y4 = 49.45000000 \sin(z4) + 50.45000000 \right] \tag{4.10}$$

```
M2 > Phi2:=Transformation(M2,M1,eqn2):
```

Here is the metric in our new coordinates.

```
M2 > gM2:=Pullback(Phi2,gM1);
```

$$gM2 := 225 \cos(z1)^2\, dz1 \otimes dz1 + \frac{81\, (-81 \cos(z2)^2 + 198 \sin(z2) + 203) \cos(z2)^2}{4}\, dz2 \otimes dz2 - 222.5250000\, (9 \sin(z2) + 11) \cos(z2) \cos(z4)\, dz2 \otimes dz4 + 225 \cos(z3)^2\, dz3 \otimes dz3 - 222.5250000\, (9 \sin(z2) + 11) \cos(z2) \cos(z4)\, dz4 \otimes dz2 + 2445.302500 \cos(z4)^2\, dz4 \otimes dz4 \tag{4.11}$$

```
M2 > Gamma2:=Christoffel(gM2):
```

Here is the function in our new coordinates.

```
M2 > F2:=eval(F1,eqn2);
```

$$F2 := \left(8.6100000\, 10^5 \sqrt{15 \sin(z1) + 35} \left(\frac{9}{2} \sin(z2) + \frac{11}{2} \right) \right) \Big/ \tag{4.12}$$

$$\left(\sqrt{15\sin(z3)+35}\right.$$
$$\sqrt{\left(\frac{9}{2}\sin(z2)+\frac{11}{2}\right)^{2}-49.45000000\sin(z4)-50.45000000}\Bigg)$$
$$+5.5350000\,10^{5}\sin(z3)+1.29150000\,10^{6}$$
$$+\frac{7.720000000\,10^{8}\left(\frac{9}{2}\sin(z2)+\frac{11}{2}\right)^{0.219}}{15\sin(z1)+35}-\frac{7.654300000\,10^{8}}{15\sin(z1)+35}$$

We have to be careful to not initialize at a point which doesn't satisfy our x2 constraints.

```
M2 > mypoint:=[Pi/4,Pi/4,Pi/4,Pi/6];
```

$$mypoint := \left[\frac{1}{4}\pi, \frac{1}{4}\pi, \frac{1}{4}\pi, \frac{1}{6}\pi\right] \tag{4.13}$$

Now we optimize. The learning rate is very small, since the function and its gradients are very large.

```
M2 > Mpath:=myGradDesPath(F2,mypoint,0.0000001,43,gM2,Gamma2)
:
```

Here we reiterate the mapping to get back to the original points.

```
M2 > evals2:=[seq(rhs(eqn2[i]),i=1..nops(eqn2))];
```

$$evals2 := \left[15\sin(z1)+35, \frac{9}{2}\sin(z2)+\frac{11}{2}, 15\sin(z3)+35,\right.$$
$$\left.49.45000000\sin(z4)+50.45000000\right] \tag{4.14}$$

```
M2 > evals1:=[seq(rhs(eqn1[i]),i=1..nops(eqn1))];
```

$$evals1 := [y1, y2, y3, y2^{2}-y4] \tag{4.15}$$

Now we look to see our path in the original coordinates.

```
M2 > Rpath1:=[seq(evalf(eval(evals2,ListTools:-FlattenOnce
(convert([z1,z2,z3,z4]=Mpath[i],listofequations)))),i=1.
.nops(Mpath))]:
M2 > Rpath2:=[seq(evalf(eval(evals1,ListTools:-FlattenOnce
(convert([y1,y2,y3,y4]=Rpath1[i],listofequations)))),i=
1..nops(Rpath1))]:
```

Were our x2 conditions violated?

```
M2 > seq(Rpath2[i][2],i=1..nops(Rpath2));
```

8.681980514, 8.42164110512262, 8.30962349562597, 8.19709225282217, 8.08401801873330, 7.97037079181963, 7.85611979930466, 7.74123336602703, 7.62567877808764, 7.50942213946538, 7.39242821960942, 7.27466028875343, 7.15607993777218, 7.03664689031198, 6.91631876654599, 6.79505087420784, 6.67279576540936, 6.54950267358897, 6.42511534178014, 6.29956193329854, 6.17285220739153, 6.04485869584328, 5.91551977483746, 5.78476090996711, 5.65250065127756, 5.51865035576602, 5.38311324397534, 5.24578325082799, 5.10654369033745, 4.96526555881291, 4.82180521720519, 4.67599571876213, 4.52770277610351, 4.37670569726911, 4.22276640435043, 4.06563066413528, 3.90500237685956, 3.74053713586947, 3.57183101380168, 3.39840511930772, 3.21968441475541, 3.03496810387440, 2.84338713932151, 2.64384129586379 **(4.16)**

They were not. Let's check how close we got to the boundary curve.

```
M2 > mycurve:=proc(x,y)
       return x^2-y;
       end:
```

Did we ever cross it?

```
M2 > [seq(mycurve(Rpath2[i][2],Rpath2[i][4]),i=1..nops
       (Rpath2))];
```

[75.17500000, 66.6883492529636, 64.7725745001729, 62.8746163570125, 60.9943041036721, 59.1314883472470, 57.2860397669251, 55.4578480500853, 53.6468209983370, 51.8528837863793, 50.0759783597930, 48.3160629421524, 46.5731116561181, 44.8471144170560, 43.1380767128816, 41.4460197468333, 39.7709801314155, 38.1130103675491, 36.4721790304316, 34.8485730668540, 33.2422688312940, 31.6534325339165, 30.0821928300667, 28.5287060486610, 26.9931508621359, 25.4757293493533, 23.9766683427587, 22.4962213651981, 21.0346719068107, 19.5923354613142, 18.1695635792359, 16.7667487957343, 15.3843227126764, 14.0227768972397, 12.6826677384408, 11.3646091077151, 10.0692970855723, 8.79752319095269, 7.55019331367823, **(4.17)**

6.32835264033067, 5.13321890762486, 3.96622759295972,
2.82909392370053, 1.72390932416370]

No we didn't. Let's see how the function has decreased.

```
M2 > myloss:=proc(p,F)
       return eval(F,ListTools:-FlattenOnce(convert([x1,x2,x3,
       x4]=p,listofequations))) #(p[1]-q[1])^2 + (p[2]-q[2])^2
       + (p[3]-q[3])^2 + (p[4]-q[4])^2;
       end:
M2 > losses:=[seq(evalf(myloss(Rpath2[i],F)), i=1..nops
       (Rpath2))]:
M2 > plot([seq(i,i=1..nops(Rpath2))],losses);
```



Here's where we ended.

```
M2 > Rpath2[-1];
```

[46.2296618310619, 2.64384129586379, 45.5732125069145, **(4.18)**

5.26598747355103]

Not too bad. But we're really close to the curvy boundary: now let's try optimizing using an equality constraint.

```
M2 > DGEnvironment[Manifold]([u1,u2,u3],T1);
```

$$\text{Manifold: } T1 \tag{4.19}$$

Here is the mapping we'll use to pull back the metric.

```
T1 > eqn1a:=[x1=u1,x2=u2,x3=u3,x4=u2^2-1];
```

$$eqn1a := \left[x1 = u1,\ x2 = u2,\ x3 = u3,\ x4 = u2^2 - 1\right] \tag{4.20}$$

```
T1 > Theta1:=Transformation(T1,R4,eqn1a);
```

$$\Theta 1 := x1 = u1,\ \ x2 = u2,\ \ x3 = u3,\ \ x4 = u2^2 - 1 \tag{4.21}$$

Here is the metric in our new coordinates.

```
T1 > gT1:=Pullback(Theta1,gR4);
```

$$gT1 := du1 \otimes du1 + \left(4\, u2^2 + 1\right) du2 \otimes du2 + du3 \otimes du3 \tag{4.22}$$

Here is the function in our new coordinates.

```
T1 > Ft1:=eval(F,eqn1a);
```

$$Ft1 := \frac{8.6100000\, 10^5 \sqrt{u1}\ u2}{\sqrt{u3}\, \sqrt{u2^2 - 1}} + 36900.00\, u3 + \frac{7.720000000\, 10^8\, u2^{0.219}}{u1} - \frac{7.654300000\, 10^8}{u1} \tag{4.23}$$

```
T1 > DGEnvironment[Manifold]([v1,v2,v3],T2);
```

$$\text{Manifold: } T2 \tag{4.24}$$

Careful: now we have sqrt(1.1) <= u2 <= sqrt(61), since we're on the curve. We'll change coordinates to ensure our rectangular inequalities are met.

```
T2 > eqn2a:=[u1=15*sin(v1)+35, u2=(sqrt(61)-sqrt(1.1))/2*sin
     (v2)+abs(sqrt(61)-(sqrt(61)-sqrt(1.1))/2), u3=15*sin(v3)
     +35];
```

$$eqn2a := \left[u1 = 15 \sin(v1) + 35,\ u2 = \frac{1}{2}\left(\sqrt{61} - 1.048808848\right) \sin(v2) + \frac{1}{2}\sqrt{61} + 0.5244044240,\ u3 = 15 \sin(v3) + 35\right] \tag{4.25}$$

```
T2 > Theta2:=Transformation(T2,T1,eqn2a);
```

(4.26)

$$\Theta 2 := u1 = 15\sin(v1) + 35,\ \ u2 = \frac{\left(\sqrt{61} - 1.048808848\right)\sin(v2)}{2} + \frac{\sqrt{61}}{2} + 0.5244044240,\ \ u3 = 15\sin(v3) + 35 \tag{4.26}$$

Here's the metric in our new coordinates.

```
T2 > gT2:=Pullback(Theta2,gT1);
```

$$gT2 := 225\cos(v1)^2\, dv1 \otimes dv1 + \left(1369.226608\sin(v2) - 522.5128984\cos(v2)^2 + 1430.944669\right)\cos(v2)^2\, dv2 \otimes dv2 + 225\cos(v3)^2\, dv3 \otimes dv3 \tag{4.27}$$

```
T2 > GammaT2:=Christoffel(gT2):
```

Here's the function in our new coordinates.

```
T2 > Ft2:=eval(Ft1,eqn2a);
```

$$Ft2 := \left(8.6100000\,10^5\sqrt{15\sin(v1)+35}\left(\frac{1}{2}\left(\sqrt{61}-1.048808848\right)\sin(v2) + \frac{1}{2}\sqrt{61} + 0.5244044240\right)\right) \Bigg/ \left(\sqrt{15\sin(v3)+35}\sqrt{\left(\frac{1}{2}\left(\sqrt{61}-1.048808848\right)\sin(v2) + \frac{1}{2}\sqrt{61} + 0.5244044240\right)^2 - 1}\right) + 5.5350000\,10^5\sin(v3) + 1.29150000\,10^6 + \frac{1}{15\sin(v1)+35}\left(7.720000000\,10^8\left(\frac{1}{2}\left(\sqrt{61} - 1.048808848\right)\sin(v2) + \frac{1}{2}\sqrt{61} + 0.5244044240\right)^{0.219}\right) - \frac{7.654300000\,10^8}{15\sin(v1)+35} \tag{4.28}$$

We'll pick up where we left off...

```
T2 > mypoint2:=[Mpath[-1][1],Mpath[-1][2],Mpath[-1][3]];
```

$$mypoint2 := [0.846014559632901,\ -0.687622741625842,\ 0.782255132548784] \tag{4.29}$$

Now we optimize. Again, we'll convert our points back to the original coordinates.

```
T2 > Mpath2:=myGradDesPath(Ft2,mypoint2,0.0000001,100,gT2,
      GammaT2):
T2 > evals2a:=[seq(rhs(eqn2a[i]),i=1..nops(eqn2a))]:
T2 > evals1a:=[seq(rhs(eqn1a[i]),i=1..nops(eqn1a))]:
T2 > Rpath1a:=[seq(evalf(eval(evals2a,ListTools:-FlattenOnce
      (convert([v1,v2,v3]=Mpath2[i],listofequations)))),i=1..
      nops(Mpath2))]:
T2 > Rpath2a:=[seq(evalf(eval(evals1a,ListTools:-FlattenOnce
      (convert([u1,u2,u3]=Rpath1a[i],listofequations)))),i=1..
      nops(Rpath1a))]:
```

Did we ever leave the curve?

```
T2 > [seq(mycurve(Rpath2a[i][2],Rpath2a[i][4]),i=1..nops
      (Rpath2a))];
```

[1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., **(4.30)**
1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1.,
1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1.,
1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1., 1.,
1., 1., 1., 1.]

What does the loss look like?

```
T2 > losses2:=[seq(evalf(myloss(Rpath2a[i],F)), i=1..nops
      (Rpath2a))]:
T2 > plot([seq(i,i=1..nops(Rpath2a))],losses2);
```



Starting loss (from the problem onset):

T2 > losses[1]

$$2.871406358\ 10^{7} \tag{4.31}$$

loss when equality took over:

T2 > losses[-1]

$$6.78532561304750\ 10^{6} \tag{4.32}$$

Loss at the end:

T2 > losses2[-1]

$$4.06737956182513\ 10^{6} \tag{4.33}$$

Total optimization algorithm:

```
T2 > plot([seq(i,i=1..nops(Rpath2)+nops(Rpath2a))],
     ListTools:-FlattenOnce([losses,losses2]));
```



Example 4: a contrived example

This is a 3-d example.

```
> DGEnvironment[Manifold]([x,y,z],R3);
```

Manifold: R3 **(5.1)**

```
R3 > gR3:=evalDG(dx &t dx + dy &t dy + dz &t dz);
```

(5.2)

$$gR3 := dx \otimes dx + dy \otimes dy + dz \otimes dz \quad \textbf{(5.2)}$$

Here is the (simple) loss function.

```
R3 > F:=x^2+y^2+(z-Pi/2)^2;
```

$$F := x^2 + y^2 + \left(z - \frac{1}{2}\pi\right)^2 \quad \textbf{(5.3)}$$

The constraint is that the following function must be 0.

```
R3 > f1:=(x+z)/sqrt(x^2+y^2+z^2) - sqrt(2)*sin(Pi/3*sin(sqrt
     (x^2+y^2+z^2)));
```

$$f1 := \frac{x+z}{\sqrt{x^2+y^2+z^2}} - \sqrt{2}\,\sin\left(\frac{1}{3}\pi\sin\left(\sqrt{x^2+y^2+z^2}\right)\right) \quad \textbf{(5.4)}$$

Here's the minimum function value everywhere:

```
R3 > eval(F,[x=0,y=0,z=Pi/2]);
```

$$0 \quad \textbf{(5.5)}$$

But the point (0,0,pi/2) doesn't satisfy our constraint.

```
R3 > simplify(eval(f1,[x=0,y=0,z=Pi/2]));
```

$$1 - \frac{1}{2}\sqrt{2}\,\sqrt{3} \quad \textbf{(5.6)}$$

That means we have work to do.

It's going to be hard to get meaningful vector fields the usual way. Let's see if one is a linear combination of Killing vectors.

```
R3 > kv:=KillingVectors(gR3);
```

$$kv := \left[-z\,\partial_x + x\,\partial_z,\ -z\,\partial_y + y\,\partial_z,\ \partial_z,\ -y\,\partial_x + x\,\partial_y,\ \partial_y,\ \partial_x\right] \quad \textbf{(5.7)}$$

```
R3 > j1:=collect(LieDerivative(evalDG(a*kv[1]+b*kv[2]+c*kv[3]
     +d*kv[4]+e*kv[5]+f*kv[6]),f1),{x,y,z});
```

$$j1 := -\frac{f x^2}{(x^2+y^2+z^2)^{3/2}} + \left(-\frac{e\,y}{(x^2+y^2+z^2)^{3/2}} + \left(-\frac{f}{(x^2+y^2+z^2)^{3/2}}\right.\right. \quad \textbf{(5.8)}$$

$$\left.- \frac{c}{(x^2+y^2+z^2)^{3/2}}\right) z$$

$$- \frac{1}{3} \frac{f\sqrt{2}\,\pi\cos\left(\sqrt{x^2+y^2+z^2}\right)\cos\left(\frac{1}{3}\pi\sin\left(\sqrt{x^2+y^2+z^2}\right)\right)}{\sqrt{x^2+y^2+z^2}}$$

$$+ \frac{a}{\sqrt{x^2+y^2+z^2}}\Bigg) x + \Bigg(- \frac{e\,z}{(x^2+y^2+z^2)^{3/2}} - \frac{d}{\sqrt{x^2+y^2+z^2}}$$

$$- \frac{1}{3} \frac{e\sqrt{2}\,\pi\cos\left(\sqrt{x^2+y^2+z^2}\right)\cos\left(\frac{1}{3}\pi\sin\left(\sqrt{x^2+y^2+z^2}\right)\right)}{\sqrt{x^2+y^2+z^2}}$$

$$+ \frac{b}{\sqrt{x^2+y^2+z^2}}\Bigg) y - \frac{c\,z^2}{(x^2+y^2+z^2)^{3/2}} + \Bigg(- \frac{a}{\sqrt{x^2+y^2+z^2}}$$

$$- \frac{1}{3} \frac{c\sqrt{2}\,\pi\cos\left(\sqrt{x^2+y^2+z^2}\right)\cos\left(\frac{1}{3}\pi\sin\left(\sqrt{x^2+y^2+z^2}\right)\right)}{\sqrt{x^2+y^2+z^2}}\Bigg) z$$

$$+ \frac{f}{\sqrt{x^2+y^2+z^2}} + \frac{c}{\sqrt{x^2+y^2+z^2}}$$

Now we generate 6 linear equations by evaluating this sum at 6 points.

```
R3 > eqns:=[evalf(eval(j1,[x=1,y=0,z=1])),evalf(eval(j1,[x=0,
     y=1,z=1])),evalf(eval(j1,[x=1,y=2,z=0])),evalf(eval(j1,
     [x=1,y=0,z=0])),
     evalf(eval(j1,[x=0,y=1,z=0])),evalf(eval(j1,[x=0,y=0,z=
     1]))];
```

$$eqns := [-0.08345704342\,f - 0.08345704342\,c,\ -0.4370104339\,e - 0.7071067810\,d + 0.7071067810\,b + 0.2700963471\,c - 0.7071067810\,a + 0.7071067810\,f,\ 0.6355165145\,f + 0.3766058381\,e + 0.4472135954\,a - 0.8944271908\,d + 0.8944271908\,b + 0.4472135954\,c,\ -0.5090950783\,f + a + c,\ -1.\,d - 0.5090950783\,e + b + f + c,\ -1.\,a - 0.5090950783\,c + f] \quad \textbf{(5.9)}$$

Then we solve.

```
R3 > solve(eqns);
```

$$\{a = 0.,\ b = d,\ c = 0.,\ d = d,\ e = 0.,\ f = 0.\} \quad \textbf{(5.10)}$$

This means that the following vector field is invariant:

```
R3 > X:=evalDG(kv[2]+kv[4]);
```

$$X := -y\partial_x + (-z+x)\,\partial_y + y\partial_z \tag{5.11}$$

Double-check that X(f1) = 0.

```
R3 > simplify(LieDerivative(X,f1));
```

$$0 \tag{5.12}$$

Now let's get a coordinate associated with this vector field.

```
R3 > pdsolve(LieDerivative(X,f(x,y,z))=1);
```

$$f(x, y, z) = -\frac{1}{2}\sqrt{2}\arctan\left(\frac{\left(\frac{1}{2}x - \frac{1}{2}z\right)\sqrt{2}}{\sqrt{y^2}}\right) + \_F1\left(x+z, -2\,(x+z)\,x + 2\,x^2 + y^2\right) \tag{5.13}$$

Double-check that the nonhomogeneous part will work:

```
R3 > simplify(LieDerivative(X, -sqrt(2)/2*arctan((x-z)/(sqrt
     (2)*y)) ));
```

$$1 \tag{5.14}$$

Here are a couple of options for another coordinate.

```
R3 > simplify(LieDerivative(X,sqrt(x^2+y^2+z^2)));
```

$$0 \tag{5.15}$$

```
R3 > simplify(LieDerivative(X,x+z));
```

$$0 \tag{5.16}$$

We'll choose sqrt(x^2+y^2+z^2), since that appears more prominently in f1. the other coordinate is f1 itself.

```
R3 > eqns1:=[t=-sqrt(2)/2*arctan((x-z)/(sqrt(2)*y)), u=sqrt
     (x^2+y^2+z^2), v=f1];
```

$$eqns1 := \left[t = -\frac{1}{2}\sqrt{2}\arctan\left(\frac{1}{2}\,\frac{(-z+x)\sqrt{2}}{y}\right), u = \sqrt{x^2+y^2+z^2}, v = \frac{x+z}{\sqrt{x^2+y^2+z^2}} - \sqrt{2}\sin\left(\frac{1}{3}\pi\sin\left(\sqrt{x^2+y^2+z^2}\right)\right)\right] \tag{5.17}$$

Define our new coordinate environment:

```
R3 > DGEnvironment[Manifold]([t,u,v],M);
```

(5.18)

$$\text{Manifold: } M \tag{5.18}$$

```
M > Phi:=Transformation(R3,M,eqns1);
```

$$\Phi := t = -\frac{\sqrt{2}\arctan\left(\frac{(-z+x)\sqrt{2}}{2\,y}\right)}{2},\ u = \sqrt{x^2+y^2+z^2},\ v = \frac{x+z}{\sqrt{x^2+y^2+z^2}} - \sqrt{2}\,\sin\left(\frac{\pi\sin\left(\sqrt{x^2+y^2+z^2}\right)}{3}\right) \tag{5.19}$$

But we need the inverse.

```
R3 > Phiin:=InverseTransformation(Phi);
```

$$\begin{aligned} Phiin := x = -\frac{1}{2}\Bigg(u\Bigg(\sqrt{2}\tan\left(t\sqrt{2}\right) RootOf\Bigg(\left(2\tan\left(t\sqrt{2}\right)^2+2\right)\_Z^2 \\ +2\sqrt{2}\,v\sin\left(\frac{\pi\sin\left(\sqrt{u^2}\right)}{3}\right)+v^2-2\cos\left(\frac{\pi\sin\left(\sqrt{u^2}\right)}{3}\right)^2\Bigg) \\ -\sqrt{2}\,\sin\left(\frac{\pi\sin\left(\sqrt{u^2}\right)}{3}\right)-v\Bigg)\Bigg),\ y = RootOf\Bigg(\left(2\tan\left(t\sqrt{2}\right)^2+2\right)\_Z^2 \\ +2\sqrt{2}\,v\sin\left(\frac{\pi\sin\left(\sqrt{u^2}\right)}{3}\right)+v^2-2\cos\left(\frac{\pi\sin\left(\sqrt{u^2}\right)}{3}\right)^2\Bigg)\,u,\ z \\ = \frac{1}{2}\Bigg(u\Bigg(\sqrt{2}\tan\left(t\sqrt{2}\right) RootOf\Bigg(\left(2\tan\left(t\sqrt{2}\right)^2+2\right)\_Z^2 \\ +2\sqrt{2}\,v\sin\left(\frac{\pi\sin\left(\sqrt{u^2}\right)}{3}\right)+v^2-2\cos\left(\frac{\pi\sin\left(\sqrt{u^2}\right)}{3}\right)^2\Bigg) \\ +\sqrt{2}\,\sin\left(\frac{\pi\sin\left(\sqrt{u^2}\right)}{3}\right)+v\Bigg)\Bigg) \end{aligned} \tag{5.20}$$

We'll have to coax a more explicit result out of Maple.

```
M > PhiinA:=simplify(convert(Phiin,radical),useassumptions)
      assuming u::positive;
```

$$PhiinA := x = \tag{5.21}$$

$$-\frac{1}{2}\Bigg(u\Bigg(\sqrt{2}\tan\left(t\sqrt{2}\right)$$

$$\sqrt{-\frac{2\sqrt{2}\,v\sin\left(\frac{\pi\sin(u)}{3}\right)-2\cos\left(\frac{\pi\sin(u)}{3}\right)^2+v^2}{2\tan\left(t\sqrt{2}\right)^2+2}}$$

$$-\sqrt{2}\,\sin\left(\frac{\pi\sin\left(\sqrt{u^2}\right)}{3}\right)-v\Bigg)\Bigg),\ y$$

$$=\sqrt{-\frac{2\sqrt{2}\,v\sin\left(\frac{\pi\sin(u)}{3}\right)-2\cos\left(\frac{\pi\sin(u)}{3}\right)^2+v^2}{2\tan\left(t\sqrt{2}\right)^2+2}}\ u,\ z$$

$$=\frac{1}{2}\Bigg(u\Bigg(\sqrt{2}\,\tan\left(t\sqrt{2}\right)$$

$$\sqrt{-\frac{2\sqrt{2}\,v\sin\left(\frac{\pi\sin(u)}{3}\right)-2\cos\left(\frac{\pi\sin(u)}{3}\right)^2+v^2}{2\tan\left(t\sqrt{2}\right)^2+2}}$$

$$+\sqrt{2}\,\sin\left(\frac{\pi\sin\left(\sqrt{u^2}\right)}{3}\right)+v\Bigg)\Bigg)$$

The resulting metric is a bit rough.

```
M > g:=Pullback(Phiin,gR3):
M > g1:=simplify(g,useassumptions) assuming u::positive;
```

$$g1 := -u^2\left(2\sqrt{2}\,v\sin\left(\frac{\pi\sin(u)}{3}\right)-2\cos\left(\frac{\pi\sin(u)}{3}\right)^2+v^2\right)dt\otimes dt-\Bigg(9\,v^6$$

$$+144\sin\left(\frac{\pi\sin(u)}{3}\right)\sqrt{2}\,v^3+54\sin\left(\frac{\pi\sin(u)}{3}\right)\sqrt{2}\,v^5$$

$$-8\cos\left(\frac{\pi\sin(u)}{3}\right)^6\pi^2\cos(u)^2u^2+24\cos\left(\frac{\pi\sin(u)}{3}\right)^4\pi^2\cos(u)^2u^2v^2$$

$$-2\cos\left(\frac{\pi\sin(u)}{3}\right)^2\pi^2\cos(u)^2u^2v^4-16\cos\left(\frac{\pi\sin(u)}{3}\right)^2\pi^2\cos(u)^2u^2v^2$$

$$-360\cos\left(\frac{\pi\sin(u)}{3}\right)^2\sin\left(\frac{\pi\sin(u)}{3}\right)\sqrt{2}\,v^3$$

$$+216\sin\left(\frac{\pi\sin(u)}{3}\right)\cos\left(\frac{\pi\sin(u)}{3}\right)^4\sqrt{2}\,v$$

$$-8\sin\left(\frac{\pi\sin(u)}{3}\right)\cos\left(\frac{\pi\sin(u)}{3}\right)^2\pi^2\sqrt{2}\cos(u)^2u^2v^3$$

$$-72\cos\left(\frac{\pi\sin(u)}{3}\right)^6$$

$$+16\sin\left(\frac{\pi\sin(u)}{3}\right)\cos\left(\frac{\pi\sin(u)}{3}\right)^4\pi^2\sqrt{2}\cos(u)^2u^2v$$

$$-270\cos\left(\frac{\pi\sin(u)}{3}\right)^2v^4+540\cos\left(\frac{\pi\sin(u)}{3}\right)^4v^2$$

$$-432\cos\left(\frac{\pi\sin(u)}{3}\right)^2v^2+216v^4\Bigg)\Bigg/$$

$$\Bigg(9\Bigg(40\cos\left(\frac{\pi\sin(u)}{3}\right)^2\sin\left(\frac{\pi\sin(u)}{3}\right)\sqrt{2}v^3-60\cos\left(\frac{\pi\sin(u)}{3}\right)^4v^2$$

$$+30\cos\left(\frac{\pi\sin(u)}{3}\right)^2v^4-24\sin\left(\frac{\pi\sin(u)}{3}\right)\cos\left(\frac{\pi\sin(u)}{3}\right)^4\sqrt{2}v$$

$$-6\sin\left(\frac{\pi\sin(u)}{3}\right)\sqrt{2}v^5-16\sin\left(\frac{\pi\sin(u)}{3}\right)\sqrt{2}v^3+8\cos\left(\frac{\pi\sin(u)}{3}\right)^6$$

$$+48\cos\left(\frac{\pi\sin(u)}{3}\right)^2v^2-24v^4-v^6\Bigg)\Bigg)du\otimes du$$

$$+\Bigg(u^2\cos\left(\frac{\pi\sin(u)}{3}\right)\pi\cos(u)\Bigg(-12\cos\left(\frac{\pi\sin(u)}{3}\right)^2\sqrt{2}v^2-16\sin\left(\frac{\pi\sin(u)}{3}\right)\cos\left(\frac{\pi\sin}{3}\right.$$

$$\Bigg(3\Bigg(40\cos\left(\frac{\pi\sin(u)}{3}\right)^2\sin\left(\frac{\pi\sin(u)}{3}\right)\sqrt{2}v^3-60\cos\left(\frac{\pi\sin(u)}{3}\right)^4v^2$$

$$+30\cos\left(\frac{\pi\sin(u)}{3}\right)^2v^4-24\sin\left(\frac{\pi\sin(u)}{3}\right)\cos\left(\frac{\pi\sin(u)}{3}\right)^4\sqrt{2}v$$

$$-6\sin\left(\frac{\pi\sin(u)}{3}\right)\sqrt{2}v^5-16\sin\left(\frac{\pi\sin(u)}{3}\right)\sqrt{2}v^3+8\cos\left(\frac{\pi\sin(u)}{3}\right)^6$$

$$+48\cos\left(\frac{\pi\sin(u)}{3}\right)^2v^2-24v^4-v^6\Bigg)\Bigg)du\otimes dv$$

$$+\Bigg(u^2\cos\left(\frac{\pi\sin(u)}{3}\right)\pi\cos(u)\Bigg(-12\cos\left(\frac{\pi\sin(u)}{3}\right)^2\sqrt{2}v^2-16\sin\left(\frac{\pi\sin(u)}{3}\right)\cos\left(\frac{\pi\sin}{3}\right.$$

$$\Bigg(3\Bigg(40\cos\left(\frac{\pi\sin(u)}{3}\right)^2\sin\left(\frac{\pi\sin(u)}{3}\right)\sqrt{2}v^3-60\cos\left(\frac{\pi\sin(u)}{3}\right)^4v^2$$

$$+30\cos\left(\frac{\pi\sin(u)}{3}\right)^2v^4-24\sin\left(\frac{\pi\sin(u)}{3}\right)\cos\left(\frac{\pi\sin(u)}{3}\right)^4\sqrt{2}v$$

$$-6\sin\left(\frac{\pi\sin(u)}{3}\right)\sqrt{2}v^5-16\sin\left(\frac{\pi\sin(u)}{3}\right)\sqrt{2}v^3+8\cos\left(\frac{\pi\sin(u)}{3}\right)^6$$

$$+48\cos\left(\frac{\pi\sin(u)}{3}\right)^2 v^2-24v^4-v^6\Big)\Big)\,dv\otimes du+\Big(u^2\Big($$

$$-8\sin\left(\frac{\pi\sin(u)}{3}\right)\cos\left(\frac{\pi\sin(u)}{3}\right)^2\sqrt{2}\,v+4\sin\left(\frac{\pi\sin(u)}{3}\right)\sqrt{2}\,v^3$$

$$-12\cos\left(\frac{\pi\sin(u)}{3}\right)^2 v^2+4\cos\left(\frac{\pi\sin(u)}{3}\right)^4+v^4+8v^2\Big)\Big)\Big/$$

$$\Big(40\cos\left(\frac{\pi\sin(u)}{3}\right)^2\sin\left(\frac{\pi\sin(u)}{3}\right)\sqrt{2}\,v^3-60\cos\left(\frac{\pi\sin(u)}{3}\right)^4 v^2$$

$$+30\cos\left(\frac{\pi\sin(u)}{3}\right)^2 v^4-24\sin\left(\frac{\pi\sin(u)}{3}\right)\cos\left(\frac{\pi\sin(u)}{3}\right)^4\sqrt{2}\,v$$

$$-6\sin\left(\frac{\pi\sin(u)}{3}\right)\sqrt{2}\,v^5-16\sin\left(\frac{\pi\sin(u)}{3}\right)\sqrt{2}\,v^3+8\cos\left(\frac{\pi\sin(u)}{3}\right)^6$$

$$+48\cos\left(\frac{\pi\sin(u)}{3}\right)^2 v^2-24v^4-v^6\Big)\,dv\otimes dv$$

M > Gamma1:=Christoffel(g1):

Here's the function in our new coordinates. We'll have to get a more explicit form later.

M > Fm:=simplify(Pullback(Phiin,F),useassumptions) assuming u::positive;

$$Fm:= \tag{5.23}$$

$$-\frac{1}{4}\,\frac{1}{\cos(t\sqrt{2})}\Big(2\sin(t\sqrt{2})\,RootOf\Big(2\sqrt{2}\,v\sin\left(\frac{1}{3}\pi\sin(u)\right)\cos(t\sqrt{2})^2$$

$$-2\cos\left(\frac{1}{3}\pi\sin(u)\right)^2\cos(t\sqrt{2})^2+v^2\cos(t\sqrt{2})^2+2\_Z^2\Big)\sqrt{2}\,u\pi$$

$$+2\sin\left(\frac{1}{3}\pi\sin(u)\right)\sqrt{2}\,u\pi\cos(t\sqrt{2})+2uv\pi\cos(t\sqrt{2})$$

$$-\pi^2\cos(t\sqrt{2})-4u^2\cos(t\sqrt{2})\Big)$$

Now remember: v=0.

M > DGEnvironment[Manifold]([tau,mu],M2);

$$\text{Manifold: } M2 \tag{5.24}$$

M2 > Phi2:=Transformation(M2,M,[t=tau,u=mu,v=0]);

$$\Phi 2 := t=\tau,\ u=\mu,\ v=0 \tag{5.25}$$

Here's the metric on our new coordinates. not nearly as bad looking.

```
M2 > g2:=Pullback(Phi2,g1);
```

$$g2 := 2\,\mu^2 \cos\left(\frac{\pi \sin(\mu)}{3}\right)^2 d\tau \otimes d\tau + \left(\frac{\pi^2 \cos(\mu)^2 \mu^2}{9} + 1\right) d\mu \otimes d\mu \quad \textbf{(5.26)}$$

The function itself is still a bit dicey, though.

```
M2 > Fm2:=convert(Pullback(Phi2,Fm),radical);
```

$$Fm2 := -\frac{1}{4}\,\frac{1}{\cos(\tau\sqrt{2})}\left(2\sin\left(\frac{1}{3}\,\pi\sin(\mu)\right)\sqrt{2}\,\mu\,\pi\cos(\tau\sqrt{2}) + 2\sin(\tau\sqrt{2})\sqrt{\cos\left(\frac{1}{3}\,\pi\sin(\mu)\right)^2\cos(\tau\sqrt{2})^2}\,\sqrt{2}\,\mu\,\pi - \pi^2\cos(\tau\sqrt{2}) - 4\,\mu^2\cos(\tau\sqrt{2})\right) \quad \textbf{(5.27)}$$

Let's coax an even more simplified expression out of Maple.

```
M2 > Fm2a:=simplify(Fm2,symbolic)
```

$$Fm2a := -\frac{1}{2}\cos\left(\frac{1}{3}\,\pi\sin(\mu)\right)\sin(\tau\sqrt{2})\,\pi\sqrt{2}\,\mu - \frac{1}{2}\sin\left(\frac{1}{3}\,\pi\sin(\mu)\right)\sqrt{2}\,\mu\,\pi + \frac{1}{4}\,\pi^2 + \mu^2 \quad \textbf{(5.28)}$$

Here are the Christoffel symbols.

```
M2 > Gamma2:=Christoffel(g2);
```

$$\Gamma 2 := \nabla_{\partial_\tau}\partial_\tau = -\frac{6\,\mu\cos\left(\frac{\pi\sin(\mu)}{3}\right)\left(-\mu\,\pi\cos(\mu)\sin\left(\frac{\pi\sin(\mu)}{3}\right) + 3\cos\left(\frac{\pi\sin(\mu)}{3}\right)\right)}{\pi^2\cos(\mu)^2\mu^2 + 9}\,\partial_\mu, \nabla_{\partial_\tau}\partial_\mu = \frac{-\mu\,\pi\cos(\mu)\sin\left(\frac{\pi\sin(\mu)}{3}\right) + 3\cos\left(\frac{\pi\sin(\mu)}{3}\right)}{3\,\mu\cos\left(\frac{\pi\sin(\mu)}{3}\right)}\,\partial_\tau, \nabla_{\partial_\mu}\partial_\tau = \frac{-\mu\,\pi\cos(\mu)\sin\left(\frac{\pi\sin(\mu)}{3}\right) + 3\cos\left(\frac{\pi\sin(\mu)}{3}\right)}{3\,\mu\cos\left(\frac{\pi\sin(\mu)}{3}\right)}\,\partial_\tau \quad \textbf{(5.29)}$$

$$, \nabla_{\partial_\mu} \partial_\mu = \frac{\pi^2 \cos(\mu)\, \mu\, (-\mu \sin(\mu) + \cos(\mu))}{\pi^2 \cos(\mu)^2 \mu^2 + 9}\, \partial_\mu$$

Now we're ready to start gradient descent.

```
M2 > initialpoint:=[0,1];
```

$$initialpoint := [0, 1] \tag{5.30}$$

```
M2 > Mpath:=myGradDesPath(Fm2a,initialpoint,0.01,400,g2,
     Gamma2):
M2 > Mpath[-1];
```

$$[1.11071656488640, 1.44187733887960] \tag{5.31}$$

```
M2 > Rpath1:=[seq(ApplyTransformation(Phi2,Mpath[i]),i=1..
     nops(Mpath))]:
M2 > Rpath2:=[seq(evalf(ApplyTransformation(PhiinA,Rpath1[i])
     ),i=1..nops(Rpath1))]:
```

Here's the ending point in the original coordinates:

```
R3 > Rpath2[-1];
```

$$[0.361068102080516, 0.0000043153476177614, 1.39593699162158] \tag{5.32}$$

Here's the loss function value at this point.

```
R3 > eval(F,[x=Rpath2[-1][1], y=Rpath2[-1][2], z=Rpath2[-1]
     [3]]);
```

$$0.160945961527631 \tag{5.33}$$

We wish to compare this value to a value which we believe is the "ground truth."

As this is a contrived example, we know how the surface is generated:

```
M2 > xt:=t*cos(Pi/3*sin(t))*cos(s);
     yt:=t*cos(Pi/3*sin(t))*sin(s);
     zt:=t*sin(Pi/3*sin(t));
```

$$xt := t \cos\left(\frac{1}{3}\pi \sin(t)\right) \cos(s)$$

$$yt := t \cos\left(\frac{1}{3}\pi \sin(t)\right) \sin(s)$$

$$zt := t \sin\left(\frac{1}{3}\pi \sin(t)\right) \tag{5.34}$$

But that's assuming the axis of rotation is the z axis. So we need to rotate the point (0,0,pi/2) about the y-axis an amount of pi/4.

Really, it's the surface itself that is rotated, not the point (0,0,pi/2). But in calculating the distance, we will pretend that the axis of revolution of the surface is the z axis and that the minimum point is rotated instead.

M2 > kv

$$\left[-z\,\partial_x + x\,\partial_z,\ -z\,\partial_y + y\,\partial_z,\ \partial_z,\ -y\,\partial_x + x\,\partial_y,\ \partial_y,\ \partial_x\right] \tag{5.35}$$

R3 > yRot:=Flow(kv[1],t);

$$yRot := x = -z\sin(t) + x\cos(t),\ y = y,\ z = z\cos(t) + x\sin(t) \tag{5.36}$$

Here's the rotated point.

R3 > Pt:=eval(ApplyTransformation(yRot,[0,0,Pi/2]),t=Pi/4);

$$Pt := \left[-\frac{1}{4}\pi\sqrt{2},\ 0,\ \frac{1}{4}\pi\sqrt{2}\right] \tag{5.37}$$

Here's the distance function: distance from the known minimum to the parameterized surface.

R3 > dist:=(xt-Pt[1])^2+(yt-Pt[2])^2+(zt-Pt[3])^2;

$$dist := \left(t\cos\left(\frac{1}{3}\pi\sin(t)\right)\cos(s) + \frac{1}{4}\pi\sqrt{2}\right)^2 + t^2\cos\left(\frac{1}{3}\pi\sin(t)\right)^2\sin(s)^2 + \left(t\sin\left(\frac{1}{3}\pi\sin(t)\right) - \frac{1}{4}\pi\sqrt{2}\right)^2 \tag{5.38}$$

We'll use Maple's built-in optimization techniques to estimate the "ground truch" distance.

R3 > with(Optimization);

$$[ImportMPS,\ Interactive,\ LPSolve,\ LSSolve,\ Maximize,\ Minimize,\ NLPSolve,\ QPSolve] \tag{5.39}$$

R3 > ans:=Minimize(dist)

$$ans := [0.160934543035327227,\ [s = 3.14159265340954,\ t = 1.44643291533958]] \tag{5.40}$$

R3 > preVec:=[eval(xt,ans[2]),eval(yt,ans[2]),eval(zt,ans[2])];

$$preVec := [-0.733323709889467,\ 1.32181715733642\ 10^{-10},\ 1.24675760077559] \tag{5.41}$$

Here's Maple's estimated minimizer:

R3 > evalf(eval(ApplyTransformation(yRot,preVec),t=-Pi/4));

$$[0.363052585840794,\ 1.32181715733642\ 10^{-10},\ 1.40012892170263] \tag{5.42}$$

Compared with ours:

```
R3 > Rpath2[-1]
```

$$[0.361068102080516, 0.0000043153476177614, 1.39593699162158] \tag{5.43}$$

Now we compare the distances (Maple's first).

```
R3 > ans[1];
     eval(F,[x=Rpath2[-1][1], y=Rpath2[-1][2], z=Rpath2[-1]
     [3]]);
```

0.160934543035327227

0.160945961527631 **(5.44)**